Date: Tue, 17 Nov 1992 14:46:00 CET
From: c.kuhtz@data-hh.zer.sub.org (Christian Kuhtz)
Reply-To: c.kuhtz@data-hh.zer.sub.org
Distribution: world
Newsgroups: de.comm.gateways
Subject: Sonderzeichen, die Erklaerung
Organization: Z-NETZ BBS Network, FRG
Content-Length: 3269    
X-Gateway: UU uzercp [UZERCP V3.99]



Hi,

fuer diejenigen die es einmal genauer wissen wollen, wo specials
erlaubt sind und wo nicht...

Wir haben der folgenden Argumentation nach keine Erlaubnis im
local-part der MsgID (also local-part@domain) specials jeglicher
Art zu verwenden.

Jedoch sind in der Domain sehr wohl eckige brackets erlaubt, die auch
bei einer speziellen Adressierungsvariante benoetigt werden (nungut,
man kommt in der Regel auch ohne aus, aber :). Pro subdomain ist
genau ein Paerchen erlaubt. Bitte: Keine Beispiele fordern :)... Aargh! :)

Ich habe nachfolgend einmal das komplette Schema aufgelistet. Wer mir
nicht glaubt, kann einmal dort nachsehen. Damit ist auch diese Frage
mehr oder weniger transparent nachvollziehbar geklaert :).

Gruss,
   Christian

PS: *(expression) meint, dass die expression in den klammern beliebig
    oft an dieser Stelle eingefuegt werden kann.
    expression / expression meint ein entweder-oder.


------------------------------------------------------------------(cut here)--


     msg-id      =  "<" addr-spec ">"            ; Unique message id

     addr-spec   =  local-part "@" domain        ; global address


     local-part  =  word *("." word)             ; uninterpreted
                                                 ; case-preserved

     domain      =  sub-domain *("." sub-domain)

     sub-domain  =  domain-ref / domain-literal

     domain-ref  =  atom                         ; symbolic reference
     
     domain-literal =  "[" *(dtext / quoted-pair) "]"

     dtext       =  <any CHAR excluding "[",     ; => may be folded
                     "]", "\" & CR, & including
                     linear-white-space>

     linear-white-space =  1*([CRLF] LWSP-char)  ; semantics = SPACE
                                                 ; CRLF => folding

     word        =  atom / quoted-string

     quoted-string = <"> *(qtext/quoted-pair) <">; Regular qtext or
                                                 ;   quoted chars.

     quoted-pair =  "\" CHAR                     ; may quote any char

     atom        =  1*<any CHAR except specials, SPACE and CTLs>

     specials    =  "(" / ")" / "<" / ">" / "@"  ; Must be in quoted-
                 /  "," / ";" / ":" / "\" / <">  ;  string, to use
                 /  "." / "[" / "]"              ;  within a word.

     CTL         =  <any ASCII control           ; (  0- 37,  0.- 31.)
                     character and DEL>          ; (    177,     127.)
     SPACE       =  <ASCII SP, space>            ; (     40,      32.)
     <">         =  <ASCII quote mark>           ; (     42,      34.)
     CHAR        =  <any ASCII character>        ; (  0-177,  0.-127.)
     LWSP-char   =  SPACE / HTAB                 ; semantics = SPACE

-----------------------------------------------------------(Anmerkung)--

Die zitierten Teile entstammen RFC822. Sie tauchen nicht in dieser
Reihenfolge im Text auf, sind eher "verstraeut" in mehren Bloecken.
Muesste aber in etwa in den ersten 10 Seiten sein.



--
Christian Kuhtz, Gruener Weg 69c, D-2000 Norderstedt, FRG voice: +49-40-5244138
ftz e.V. unix development - data: +49-40-5248447  email: chk@DATA-HH.ComLink.DE
  we use 386BSD - we're not a copystation but you can ask us about 386BSD :)
## CrossPoint v1.99 ##